

BAB III

PERANCANGAN DAN PEMBUATAN ALAT

A. Perancangan Perangkat Keras

Dalam bab ini akan dijelaskan tentang perancangan dan pembuatan seluruh perangkat keras dalam pembuatan “Perancangan Alat Ukur Kualitas Sinyal HandPhone Pada Tipe GSM”, adapun perangkat keras yang dirancang berupa rangkaian sistem minimum 89C51, regulator 5V, rangkaian LCD dan rangkaian LED.


Adapun rangkaian – rangkaian diatas dapat dilihat pada gambar di bawah ini:

1. Rangkaian Sistem Minimum 89C51

Pada rangkaian sistem minimum 89C51 diperlukan beberapa komponen antara lain :

- 1 buah chip mikrokontroler 89C51
- 1 buah kristal 22,1184 MHz
- 2 buah kapasitor 22 pF
- 1 buah resistor 10 K ohm
- 1 buah kapasitor elektrolit 1 uF/50V

Rangkaian dari sistem minimum 89C51 adalah sebagai berikut :




Gambar 3.1 Rangkain Sistem Minimum 89C51

2. Rangkaian Regulator 5V

Rangkaian ini diperlukan untuk memberi tegangan / power ke setiap rangkaian yang lain agar dapat beroperasi sesuai dengan karakteristik setiap komponen. Pada rangkaian regulator diperlukan beberapa komponen, yaitu :

- 4 buah dioda IN4004
- 1 buah kapasitor elektrolit 100 uF/25V
- 1 buah kapasitor 100 nF
- 1 buah IC LM780SCT

Rangkaian regulator yang dibuat dalam tugas akhir ini dapat dilihat pada gambar dibawah ini :




Gambar 3.2 Rangkaian Regulator 5V

3. Rangkaian LED

Rangkaian LED digunakan sebagai indikator untuk menandakan seberapa besar kualitas sinyal yang dihasilkan dengan cara menampilkan melalui nyala atau tidaknya LED. Pada rangkaian ini, terdapat 4 buah LED yang terdiri dari LED pertama (L_1), LED kedua (L_2), LED ketiga (L_3), dan LED keempat (L_4). LED tersebut akan menyala berdasarkan nilai RSSI yang diperoleh.


Rangkaian LED tersebut adalah sebagai berikut :



Gambar 3.3 Rangkaian LED

4. Rangkaian LCD

Rangkaian LCD dalam tugas akhir ini berfungsi untuk menampilkan kualitas sinyal dalam bentuk angka. LCD yang dipakai yaitu tipe 4470 buatan Hitachi dengan tipe modul yaitu 2 x 16 . Angka 16 menunjukkan kapasitas maksimal karakter yang dapat ditampilkan sementara angka 2 menunjukkan jumlah baris yang ada. Rangkaian LCD dapat dilihat dari gambar dibawah ini :




Gambar 3.4 Rangkaian LCD

5. Spesifikasi Alat

- Tegangan supply yang digunakan yaitu 220 V.
- Tegangan kerja 5 V.
- Input yang digunakan berupa HandPhone dengan tipe Siemens baik itu S25, C35, M35 atau ME45.
- Sebagai kontrol digunakan IC Mikrokontroller AT89C51.
- Tampilan output berupa Led dan LCD
- LCD yang digunakan yaitu dot matrik 4470 buatan Hitachi dengan 8 digit (4 digit satuan, 2 digit dibelakang koma).
- Software* (program) dibuat dalam bahasa assembler mikrokontroller.
- Temperatur kerja menggunakan suhu ruangan 27° C

6. Diagram Alir Alat

Untuk memudahkan dalam pembuatan alat tersebut maka dibutuhkan diagram alir yang fungsinya memudahkan seseorang dalam menuangkan ide-idenya atau pemahaman secara global, untuk diagram alir tugas akhir ini dapat dilihat seperti dibawah ini:



Gambar 3.5 Diagram Alir Alat

7. Prinsip Kerja Alat

Prinsip kerja dari alat ini yaitu Mikrokontroler mengirimkan perintah kepada HP untuk mencocokkan baudrate yang telah disimpan di memori mikrokontroler dengan baudrate yang ada di handphone. Dikarenakan baudrate pada setiap handphone berbeda-beda maka mikrokontroler mengirimkan perintah tersebut secara berurutan (*polling*). Pencocokan baudrate ini dikarenakan sebelum dapat berkomunikasi dengan handphone, harus diseting terlebih dahulu kecepatan transfer data pada handphone tersebut. Setelah terdapat kecocokan data baudrate tersebut mikrokontroler

mengirimkan perintah (*command*) berupa protokol *at command*. Protokol *at command* yang digunakan adalah AT+CSQ dikarenakan perintah tersebut digunakan untuk menampilkan kualitas sinyal yang dihasilkan. Mikrokontroler mengirimkan perintah tersebut kemudian Handphone merespon atau memberi jawaban data CSQ. Setelah mendapatkan data CSQ kemudian mikrokontroler menampilkan ke layar LCD dan LED. Pada alat yang dibuat ini menggunakan sistem *half duplex* dikarenakan lalu lintas data sepenuhnya diatur oleh mikrokontroler yang memberikan perintah sedangkan HandPhone hanya memberikan jawaban (*response*).

B. Perancangan Perangkat Lunak

Perancangan perangkat lunak/program merupakan perancangan kedua yang penulis lakukan setelah melakukan perancangan perangkat keras. Sebelum melakukan perancangan program harus membuat flowchart terlebih dahulu


Sedangkan diagram alir (*Flowchart*) program dapat dilihat dibawah ini :

1. Flowchart/Diagram Alir

Diagram alir merupakan skema yang menggambarkan urutan kegiatan suatu program dari awal sampai akhir. Kegunaan dari *flowchart* adalah bila suatu *flowchart* lengkap telah selesai dibuat, gambaran lengkap tentang proses pemikiran seorang programmer dalam memecahkan suatu masalah dapat diikuti. *Flowchart* sangat penting untuk pemeriksaan program yang telah selesai, juga dapat membantu orang lain dalam memahami algoritma yang tepat yang dibuat oleh seorang programmer. Urutan langkah-langkah dari

sebuah program ditunjukkan dengan simbol anak panah juga digunakan simbol-simbol yang lain untuk menunjukkan operasi yang dilaksanakan pada tiap-tiap langkah dari suatu program.

Diagram alir dari program perancangan Alat Ukur Kualitas Sinyal pada Tipe HandPhone GSM adalah sebagai berikut :



Gambar 3.6 Diagram Alir Program

2. Listing Program

Adapun listing program dari perancangan Alat Ukur Kekuatan Sinyal pada Tipe HandPhone GSM dan CDMA adalah sebagai berikut :

```

1.      ;-----
2.      ; USER PIN
3.      ;-----
4.      EN                Equ  0B4H ; P3.4
5.      RS                Equ  0B5H ; P3.5

```

Baris ke 1 – 5 merupakan deklarasi untuk menentukan pin out pada kaki mikrokontroler. Pada Pin alamatnya ditentukan terlebih dahulu agar dalam pengaturan ruang memori tidak terjadi penumpukkan pada alamat yang sama.

```

7.      ;-----
8.      ; USER REGISTER DEFINE
9.      ;-----
10.     Baud_Rate         Equ  008H
11.     Serial_Pointer   Equ  009H
12.     Power_Signal     Equ  00AH
13.     DELAY_MS         Equ  00BH
14.     DELAYS           Equ  00CH
15.     COUNT            Equ  00DH
17.     CMD              Equ  00EH
18.     BUFFER           Equ  00FH
19.     LED              Equ  01AH

```

Baris 7 – 19 merupakan deklarasi atau pemberian alamat pada setiap data yang di simpan dalam register. Pemberian alamat pada data tersebut ditentukan terlebih dahulu agar memudahkan dalam pengaturan register.

```

21.     ;-----
22.     ; MAIN PROGRAM
23.     ;-----
24.     ORG              0000h
25.     AJMP             Start

27.     ORG              000Bh
28.     MOV              TH0,#0F8H
29.     MOV              TLO,#030H
30.     INC              DELAY_MS
31.     RETI

34.     ORG              0100H

```

```

35.   Start:
36.   MOV     SP,#050H

```

Baris 21 – 36 merupakan program utama dari sistem pengukuran kualitas sinyal. Dimana pada baris 24 – 34 merupakan interupsi timer sedangkan baris 34 merupakan alamat awal program.

```

38.   ;-----
39.   ; INIT USER REGISTER
40.   ;-----
41.   CLR     A
42.   MOV     P0,#0H
43.   CLR     EN
44.   CLR     RS

46.   LCALL  DELAY10MS
47.   LCALL  DELAY10MS

49.   MOV     Power_Signal,A
50.   MOV     TH0,#0F8H
51.   MOV     TL0,#030H
53.   MOV     TMOD,#021H
54.   MOV     SCON,#050H      ; MODE BAUD RATE 8 BIT
55.   MOV     PCON,#80H      ; BAUD RATE 2X
56.   SETB   TR0
57.   SETB   TR1
58.   SETB   ET0
59.   SETB   EA

```

Baris ke 38 – 59 merupakan instruksi untuk mengatur interrupt timer.

```

61.   MOV     A,#250          ; delay 250 milli second
62.   LCALL  DELAYMS
63.   MOV     A,#250          ; delay 250 milli second
64.   LCALL  DELAYMS

```

Baris ke 61 – 63 merupakan instruksi untuk mengatur waktu tunda (*delay*). Waktu tunda ditentukan sebesar 250 mS.

```

66.   LCALL  INIT_LCD        ; INISIALISASI LCD
67.   MAIN:
68.   MOV     Power_Signal,A
69.   MOV     P2,#0FFH
71.   MOV     DPTR,#DISP_CONNECTING
72.   Lcall  PRINTSTRING1
74.   LCALL  BAUD_RATE_SCANNING
76.   MOV     DPTR,#DISP_CONNECTED

```

```

77.    Lcall    PRINTSTRING1
79.    MOV      COUNT,#3

```

Baris ke 66 – 79 merupakan instruksi untuk mengambil data pada HandPhone dan menampilkan pada layar LCD.

```

81.    SIGNAL?:
82.    MOV      A,#250
83.    LCALL    DELAYMS
85.    MOV      DPTR,#MESS_AT_CSQ    ; LOAD 'AT+CSQ'
86.    LCALL    SEND_CHAR_ARRAY
88.    MOV      DPTR,#MESS_OK        ; WAITING 'OK'
89.    MOV      A,#100
90.    LCALL    SERIAL_EXPECT
91.    JC       SCANNING_CSQ
93.    MOV      DPTR,#MESS_AT_CSQ?   ; LOAD 'AT+CSQ?'
94.    LCALL    SEND_CHAR_ARRAY
96.    MOV      DPTR,#MESS_OK        ; WAITING 'OK'
97.    MOV      A,#100
98.    LCALL    SERIAL_EXPECT
99.    JC       SCANNING_CSQ
101.   DJNZ     COUNT,SIGNAL?
103.   MOV      DPTR,#DISP_CONNECTION_LOST
104.   LCALL    PRINTSTRING2
105.   LCALL    Clrscr1
106.   MOV      A,#250
107.   LCALL    DELAYMS
108.   MOV      A,#250
109.   LCALL    DELAYMS
110.   LJMP     MAIN

```

Baris ke 81 – 110 merupakan instruksi untuk mengirim AT + CSQ ke HandPhone. Apabila HandPhone belum terkoneksi, mikrokontroler akan terus menerus mencari (*looping*) sampai HandPhone terkoneksi kemudian mengambil data CSQ.

```

112.   SCANNING_CSQ:
113.   MOV      COUNT,#3
114.   MOV      R1,#BUFFER
115.   MOV      DPTR,#MESS_CSQ
116.   LCALL    SCANNING_TEXT
117.   CJNE     R1,#00,CSQ_RECEVIED
118.   LJMP     MAIN
120.   CSQ_RECEVIED:
122.   LCALL    Clrscr1
124.   MOV      A,#01H

```

```

125.   LCALL   POSISI1
127.   PUSH    00
128.   MOV     A,#0DH           ; display terminated by this
character
129.   LCALL   DISP_MSG_R0
131.   POP     01
132.   MOV     DPTR,#MESS_COMMA
133.   LCALL   SCANNING_TEXT
134.   CJNE   R1,#00,CHECK_VALUES
135.   LJMP    MAIN
137. CHECK_VALUES:
138.   DEC     R1
139.   MOV     A,@R1
140.   ANL    A,#0FH
141.   DEC     R1
142.   CJNE   R1,#',PULUHAN
143.   SJMP    CHECK_VALUE
145. PULUHAN:
146.   MOV     B,A
147.   MOV     A,@R1
148.   SWAP   A
149.   ANL    A,#0F0H
150.   ORL    A,B
152. CHECK_VALUE:
153.   LCALL   DATAOUT
154.   MOV     Power_Signal,A
155.   CJNE   A,#99,DETECT
156.   MOV     A,#0
157.   SJMP    CONVERSION_LED
159. DETECT:
160.   MOV     B,#4
161.   DIV    AB
162.   INC    A
164. CONVERSION_LED:
165.   LCALL   LED_BAR
166.   MOV     P2,A
168.   LCALL   Crlscr2
169.   MOV     DPTR,#RSSI
170.   LCALL   PRINTSTRING2
171.   MOV     A,Power_Signal
172.   CJNE   A,#99,RSSI_0_31
174.   MOV     DPTR,#RSSI_99
175.   LCALL   PRINTSTRING
177.   LJMP    SIGNAL?
179. RSSI_0_31:
180.   MOV     DPTR,#RSSI_TBL

```

```

181.  MOV     B,#8
182.  MUL     AB
183.  ADD     A,DPL
184.  MOV     DPL,A
185.  CLR     A
186.  ADDC    A,DPH
187.  MOV     DPH,A
188.  LCALL   PRINTSTRING
189.  LJMP    SIGNAL?

```

Baris ke 112 – 189 merupakan instruksi pengolahan data CSQ sampai menampilkan sinyal HandPhone.

```

191.  ;-----
192.  ;PROCEDURE KIRIM TEXT
193.  ;-----
194.  SEND_CHAR_ARRAY:
195.  CLR     A
196.  MOVC    A,@A+DPTR
197.  INC     DPTR
198.  JZ      EXIT_SEND_MESS
199.  LCALL   TRANSMIT
200.  SJMP    SEND_CHAR_ARRAY
201.  EXIT_SEND_MESS:
202.  RET

```

Baris ke 191 – 202 merupakan instruksi pengiriman teks

```

204.  ;-----
205.  ;PROCEDURE KIRIM SERIAL
206.  ;-----
207.  TRANSMIT:
208.  MOV     SBUF,A
209.  JNB     TI,$
210.  CLR     TI
211.  RET

```

Baris ke 204 – 211 merupakan instruksi pengiriman data melalui serial

```

213.  ;-----
214.  ;FUNCTION SCANNING TEXT
215.  ;Input : - R1   = Text to be scanned.
216.  ;         - DPTR = Sub Text to search for.
217.  ;
218.  ;Output : - R0  = Pointer to the first occurrence of sub text DPTR in R1.
219.  ;           = Returns NULL if sub text DPTR is not found in R1.
220.  ;         - R1  = Pointer to the last occurrence of sub text DPTR in R1.
221.  ;           = Returns NULL if sub text DPTR is not found in R1.
222.  ;-----

```

```

223. SCANNING_TEXT:
224.   PUSH   DPH
225.   PUSH   DPL
227.   CLR    A
228.   MOVC   A,@A+DPTR
229.   MOV    R0,1
231. LOOP_SCAN_TEXT:
232.   XRL    A,@R1
233.   INC    R1
234.   JZ     NEXT_SCAN_TEXT?
235.   CJNE   @R1,#00,RST_SUB_TEXT
236.   LJMP   SCAN_NOT_FOUND
238. RST_SUB_TEXT:
239.   POP    DPL
240.   POP    DPH
241.   SJMP   SCANNING_TEXT
243. NEXT_SCAN_TEXT?:
244.   INC    DPTR
245.   CLR    A
246.   MOVC   A,@A+DPTR
247.   JNZ    NEXT_SCAN_TEXT
248.   DEC    R1
249.   SJMP   EXIT_SCAN_TEXT
251. NEXT_SCAN_TEXT:
252.   CJNE   @R1,#00,LOOP_SCAN_TEXT
254. SCAN_NOT_FOUND:
255.   CLR    A
256.   MOV    R1,A
257.   MOV    R0,A
259. EXIT_SCAN_TEXT:
260.   POP    DPL
261.   POP    DPH
262.   RET

```

Baris ke 213 – 216 merupakan instruksi *scanning* untuk mencari data yang sama dengan output dBm.

```

264. ;-----
265. ;FUNCTION SERIAL EXPECT
266. ;Input : - DPTR = Array of charracters (terminated by 00H) to listen for
267. ;        - Acc   = The maximum wait in milliseconds for a character
268. ;
269. ;Output : - Carry :
270. ;          1 if the expected array was received
271. ;          0 if a timeout occured before receiving the array
272. ;-----
273. SERIAL_EXPECT:

```

```

274.  PUSH    B
275.  PUSH    7
276.  PUSH    6
277.  PUSH    5
279.  CLR     RI
281.  MOV     R6,DPH
282.  MOV     R5,DPL
283.  MOV     R0,#BUFFER
285.  ADD     A,DELAY_MS
286.  MOV     R7,A
287.  CLR     A
288.  MOV     @R0,A
290.  LOOP_EXPECT:
291.  CLR     A
292.  MOVC   A,@A+DPTR
293.  SETB   C
294.  JZ     ARRAY_FOUND
295.  MOV     B,A
297.  WAIT_1CHAR:
298.  MOV     A,DELAY_MS
299.  XRL    A,R7
300.  JZ     WAS_TIMEOUT
301.  JNB    RI,WAIT_1CHAR
302.  CLR     RI
303.  MOV     A,SBUF
304.  MOV     @R0,A
305.  INC    R0
306.  MOV     @R0,#0
307.  XRL    A,B
308.  JNZ    RST_EXPECT
309.  INC    DPTR
310.  LJMP   LOOP_EXPECT
312.  RST_EXPECT:
313.  MOV     DPH,R6
314.  MOV     DPL,R5
315.  LJMP   WAIT_1CHAR
317.  WAS_TIMEOUT:
318.  CLR     C
319.  ARRAY_FOUND:
320.  POP     5
321.  POP     6
322.  POP     7
323.  POP     B
324.  RET

```

Baris ke 264 – 324 merupakan perintah proses penerimaan data + csq

```

326. ;-----
327. ; PROCEDURE TERIMA SERIAL
328. ;-----
329. RECEIVE:
330.     JNB     RI,$
331.     CLR     RI
332.     MOV     A,SBUF
333.     RET

```

Baris ke 326 – 333 merupakan instruksi penerimaan data dari serial

```

334. ;-----
335. ;PROCEDURE BAUD RATE SCANNING
336. ;-----
337. ; TH1 value:
338. ; These values are for 22.1184MHz operation
339. BAUD_RATE_TBL:
340. DB     0FFH ; 0. Baud Rate = 115200 bps
341. DB     0FEH ; 1. Baud Rate = 57600 bps
342. DB     0FDH ; 2. Baud Rate = 38400 bps
343. DB     0FAH ; 3. Baud Rate = 19200 bps
344. DB     0F4H ; 4. Baud Rate = 9600 bps
345. DB     0E8H ; 5. Baud Rate = 4800 bps
346. DB     0D0H ; 6. Baud Rate = 2400 bps
347. DB     0A0H ; 7. Baud Rate = 1200 bps
348. ;-----
349. TRY_NEW_BAUD:
350.     INC     Baud_Rate
352.     BAUD_RATE_SCANNING:
353.     LCALL   Crlscr2
354.     MOV     DPTR,#DISP_WITH
355.     Lcall   PRINTSTRING2
357.     MOV     A,Baud_Rate
358.     ANL     A,#009H
359.     MOV     B,#8
360.     MUL     AB
361.     MOV     DPTR,#DISP_BAUD_TBL
362.     ADD     A,DPL
363.     MOV     DPL,A
364.     MOV     A,B
365.     ADDC   A,DPH
366.     MOV     DPH,A
367.     Lcall   PRINTSTRING
369.     MOV     DPTR,#DISP_BPS
370.     Lcall   PRINTSTRING
372.     MOV     A,Baud_Rate
373.     ANL     A,#007H

```



```

374. CLR     TR1
375. MOV     DPTR,#BAUD_RATE_TBL
376. MOVC    A,@A+DPTR
377. MOV     TH1,A
378. MOV     TL1,A
379. SETB    TR1
381. MOV     DPTR,#MESS_AT
382. LCALL   SEND_CHAR_ARRAY
384. MOV     A,#100
385. MOV     DPTR,#MESS_OK
386. LCALL   SERIAL_EXPECT
387. JC      BAUD_MATCH
388. LJMP    TRY_NEW_BAUD
389. BAUD_MATCH:
390. RET

```

Baris ke 334 – 390 merupakan instruksi untuk mencocokkan baudrate yang telah disimpan pada mikro dengan baudrate HandPhone

```

393. ;-----
394. ; PROSEDUR INISIALISASI LCD
395. ;-----
396. INIT_LCD:
398. MOV     A,#28H
399. LCALL   CONTROLOUT
400. LCALL   DELAY_INIT_LCD
401. MOV     A,#08H
402. LCALL   CONTROLOUT
403. LCALL   DELAY_INIT_LCD
404. MOV     A,#01H
405. LCALL   CONTROLOUT
406. LCALL   DELAY_INIT_LCD
407. MOV     A,#06H
408. LCALL   CONTROLOUT
409. LCALL   DELAY_INIT_LCD
410. MOV     A,#0FH
411. LCALL   CONTROLOUT
412. LCALL   DELAY_INIT_LCD
413. MOV     A,#0CH
414. LCALL   CONTROLOUT
415. RET

```

Baris ke 393 – 415 merupakan instruksi pengkondisian awal untuk mengaktifkan

LCD

```

419. ;=====

```

```

420. ; PROCEDURE LCD
421. ;=====
422. POSISI2_1:
423.   CLR      A
424. POSISI2:
425.   ADD      A,#1100B ; BARIS KEDUA
426.   SJMP     POSISI_SUB
428. POSISI1_1:
429.   CLR      A
430. POSISI1:
431.   ADD      A,#1000B ; BARIS PERTAMA
432. POSISI_SUB:
433.   LCALL    CONTROLOUT
434.   RET
436. PRINTSTRING2:
437.   LCALL    POSISI2_1
438.   SJMP     PRINTSTRING
440. PRINTSTRING1:
441.   LCALL    POSISI1_1
443. PRINTSTRING:
444.   SJMP     OUTSTRING
445. PRINTSTRINGLOOP:
446.   LCALL    DATAOUT
447.   INC      DPTR
448. OUTSTRING:
449.   CLR      A
450.   MOVC     A,@A+DPTR
451.   JNZ      PRINTSTRINGLOOP
452.   RET
454. DISP_MSG_R0:
455.   PUSH     B
456.   MOV      B,A
458. LOOP_MSG_R0:
459.   MOV      A,@R0
460.   CJNE     A,#00,TERMINATED?
461.   POP      B
462.   RET
464. TERMINATED?:
465.   CJNE     A,B,DONT_TERMINATED
466.   POP      B
467.   RET
469. DONT_TERMINATED:
470.   LCALL    DATAOUT
471.   INC      R0
472.   SJMP     LOOP_MSG_R0

```

Baris ke 419 – 472 merupakan subrutin untuk menampilkan data dilayar LCD

```

474. ;-----
475. ;PROSEDUR KONTROL KE LCD
476. ;-----
477. CONTROLOUT:
478.  PUSH      DPH
479.  PUSH      DPL
480.  PUSH      ACC
482.  CLR       EN
483.  MOV       CMD,A
484.  ANL       A,#0F0H          ; TAKE HIGH DATA
485.  ORL       A,#0FH
486.  MOV       P0,A
487.  SETB      RS
488.  SETB      EN
489.  LCALL     DELAY
490.  CLR       EN
492.  MOV       A,CMD
493.  ANL       A,#0FH          ; TAKE LOW DATA
494.  SWAP      A
495.  ORL       A,#0FH
496.  MOV       P0,A
497.  CLR       RS
498.  SETB      EN
499.  LCALL     DELAY
500.  CLR       EN
502.  POP       ACC
503.  POP       DPL
504.  POP       DPH
505.  RET

```

Baris ke 474 – 505 merupakan subrutin pengontrolan di LCD

```

508. ;-----
509. ;PROCEDURE DATA KE LCD
510. ;-----
511. DATAOUT:
512.  PUSH      DPH
513.  PUSH      DPL
514.  PUSH      ACC
516.  MOV       CMD,A
517.  ANL       A,#0F0H          ; TAKE HIGH DATA
518.  ORL       A,#0FH
519.  MOV       P0,A
520.  CLR       RS
521.  SETB      EN
522.  LCALL     DELAY
523.  CLR       EN

```

```

525.  MOV     A,CMD
526.  ANL     A,#0FH           ; TAKE LOW DATA
527.  SWAP    A
528.  ORL     A,#0FH
529.  MOV     P0,A
530.  SETB    RS
531.  SETB    EN
532.  LCALL   DELAY
533.  CLR     EN
535.  POP     ACC
536.  POP     DPL
537.  POP     DPH
538.  RET

```

Baris ke 508 – 538 merupakan interupsi data ke LCD

```

541.  ;-----
542.  ; DISPLAY CLEAR
543.  ;-----
544.  Clrscr1:
545.  Mov     Dptr,#Clear
546.  Lcall   PRINTSTRING1
547.  Ret
549.  Clrscr2:
550.  Mov     Dptr,#Clear
551.  Lcall   PRINTSTRING2
552.  Ret

```

Baris ke 541 – 552 merupakan instruksi menghapus layar di LCD

```

554.  ;-----
555.  ;  PROCEDURE DELAY
556.  ;-----
557.  DELAY:
558.  PUSH    06H
559.  PUSH    07H
561.  MOV     R6,#04H
562.  MUTER:
563.  MOV     R7,#250
564.  DJNZ    R7,$
565.  DJNZ    R6,MUTER
567.  POP     07H
568.  POP     06H
569.  RET

```

Baris ke 554 – 569 merupakan prosedur dari waktu tunda

```

571.  ;-----
572.  ;  PROCEDURE DELAY LCD

```

```

573. ;-----
574. DELAY_INIT_LCD:
576.  PUSH    06H
577.  PUSH    07H
579.  MOV     R6,#40H
580. DLY_LCD_LP:
581.  MOV     R7,#00H
582.  DJNZ    R7,$
583.  DJNZ    R6,DLY_LCD_LP
585.  POP     07H
586.  POP     06H
587.  RET

```

Baris ke 571 – 587 merupakan prosedur waktu tunda di LCD

```

590. ;-----
591. ; PROCEDURE LED
592. ;-----
593. LED_BAR:
594.  INC     A
595.  MOVC    A,@A+PC
596.  RET
597. NYALA_LED:
598.  DB      11111111B
599.  DB      11111110B
600.  DB      11111100B
601.  DB      11111000B
602.  DB      1111B
603.  DB      1110B
604.  DB      1100B
605.  DB      1000B
606.  DB      B

```

Baris ke 590 – 606 merupakan instruksi pengaktifan LED

```

608. ;-----
609. ; delay 500 ms
610. ;-----
611. DELAY500MS:
612.  PUSH    ACC
613.  MOV     A,#32H      ; 500 milli second
615. X10MS:
616.  CALL    DELAY10MS
617.  DJNZ    ACC,X10MS
619.  POP     ACC
620.  RET
622. ;-----
623. ; DELAY 10 ms

```

```

624. ;-----
625. DELAY10MS:
626.   PUSH   ACC
627.   PUSH   00H
629.   MOV    00H,#32H
630. D10MS1:
631.   MOV    A,#0C8H
632.   DJNZ   ACC,$
633.   DJNZ   R0,D10MS1
635.   POP    00H
636.   POP    ACC
637.   RET
639. ;-----
640. ; DELAY 1 ms
641. ;-----
642. DELAY1MS:
643.   PUSH   ACC
644.   PUSH   00H
646.   MOV    00H,#05H
647. D1MS1:
648.   MOV    A,#0C8H
649.   DJNZ   ACC,$
650.   DJNZ   R0,D1MS1
652.   POP    00H
653.   POP    ACC
654.   RET
658. DELAYMS:
659.   ADD    A,DELAY_MS
660.   MOV    DELAYS,A
661. LOOP_DELAYMS:
662.   MOV    A,DELAYS
663.   XRL   A,DELAY_MS
664.   JNZ   LOOP_DELAYMS
665.   RET

```

Baris ke 608 – 665 merupakan subrutin dari waktu tunda

```

667. ;-----
668. ; DATA
669. ;----- 0123456789ABCDEF
670. RSSI_TBL:
671.   DB    '-113dBm',0H,'-111dBm',0H,'-109dBm',0H,'-107dBm',0H
672.   DB    '-105dBm',0H,'-103dBm',0H,'-101dBm',0H,'-99 dBm',0H
673.   DB    '-97 dBm',0H,'-95 dBm',0H,'-93 dBm',0H,'-91 dBm',0H
674.   DB    '-89 dBm',0H,'-87 dBm',0H,'-85 dBm',0H,'-83 dBm',0H
675.   DB    '-81 dBm',0H,'-79 dBm',0H,'-77 dBm',0H,'-75 dBm',0H
676.   DB    '-73 dBm',0H,'-71 dBm',0H,'-69 dBm',0H,'-67 dBm',0H

```

```

677.   DB   '-65 dBm',0H,'-63 dBm',0H,'-61 dBm',0H,'-59 dBm',0H
678.   DB   '-57 dBm',0H,'-55 dBm',0H,'-53 dBm',0H,'-51 dBm',0H
679.   RSSI_99:
680.   DB   'Not Known',0H
681.   RSSI:
682.   DB   'RSSI:',0H
683.   DISP_CONNECTING:
684.   DB   'Connecting.....',00h
685.   DISP_CONNECTED:
686.   DB   'Connected ',00h
687.   DISP_CONNECTION_LOST:
688.   DB   'Connection Lost',00h
689.   DISP_WITH:
690.   DB   'With ',00h
691.   DISP_BPS:
692.   DB   'Bps',00H
693.   Clear:
694.   DB   ' ',00H
695.   DISP_BAUD_TBL:
696.   DB   '115200 ',00H
697.   DB   '57600 ',00H,00H
698.   DB   '38400 ',00H,00H
699.   DB   '19200 ',00H,00H
700.   DB   '9600 ',00H,00H,00H
701.   DB   '4800 ',00H,00H,00H
702.   DB   '2400 ',00H,00H,00H
703.   DB   '1200 ',00H,00H,00H
705.   MESS_COMMA:
706.   DB   ',',00H
707.   MESS_CSQ:
708.   DB   '+CSQ:',00H
709.   MESS_AT_CSQ:
710.   DB   'AT+CSQ',0DH,00H
711.   MESS_AT_CSQ?:
712.   DB   'AT+CSQ?',0DH,00H
713.   MESS_OK:
714.   DB   'OK',0DH,00H
715.   MESS_AT:
716.   DB   'AT',0DH,00H
717.   DATA_1:
718.   DB   'AT+CSQ',0DH

```

Baris ke 667 – 718 merupakan instruksi untuk menyimpan data-data yang akan ditampilkan

